
1 [bookmark: _Toc122434485][bookmark: _Toc160457771]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D24wk12 related to the title of this CR.
Contact:	Olivier Genoud
	olivier.genoud@etsi.org
2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc160457772]Corrections required
2.1 Correction of f_MCX_SIP_Registration
	Function name
	f_MCX_SIP_Registration

	Reason for change
	Message body of SIP REGISTER needs to be optional

	Summary of change
	Message body made optional when calling cr_REGISTER_Request

	TTCN module
	MCX_SIP_Registration



	  function f_MCX_SIP_Registration(MCX_ServiceType_Type p_ServiceType) runs on MCX_IMS_PTC return IMS_DATA_REQ
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ_ByRef;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var REGISTER_Request v_RegisterReq;
    var MIME_Message v_MIME_Message;
    var template (present) SemicolonParam_List v_MandatoryContactParams := f_MCX_SIP_Register_ContactParams_RX(p_ServiceType);
    var template (present) MCX_ServiceInfo_Type v_MCS_Info;
    var MCX_MIME_XmlBodies_Type v_XmlBodies;
    
    // Step 1
    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, -, v_MandatoryContactParams);  // check REGISTER and init GlobalInfo
    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    f_IMS_Register_SecurityInit_ImsSecurity(v_RegisterReq);

    // Step 2
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));
    
    // Step 3a1, 3b1
    a_IMS_ReceiveRequest(car_IMS_Register_Request(cr_REGISTER_Request(?, ?, *, ?)), v_IMS_DATA_REQ_ByRef);
    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(v_IMS_DATA_REQ_ByRef, v_MandatoryContactParams);
    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;
    if (isvalue(v_IMS_DATA_REQ.Request.Register.messageBody.mimeMessageBody)) { /* @sic R5-240555: Step 3a1 with SIP REGISTER (access token, CSK) sic@ */
      if (not match(v_IMS_DATA_REQ.Request.Register.msgHeader.contentType, cr_ContentTypeMIME)) {
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid Content-Type");
      }
      v_MIME_Message := v_RegisterReq.messageBody.mimeMessageBody;
      f_MCX_SIP_RegistrationSubscription_DecodeMatchAndCheckMikeyMessage_SetCSK(p_ServiceType, v_MIME_Message, CSK);
      v_MCS_Info := fl_MCX_SIP_Registration_McsInfo(p_ServiceType, v_IMS_DATA_REQ.Request);
      v_XmlBodies := f_MCX_MIME_DecodeAndMatchXML(v_MIME_Message, p_ServiceType, -, v_MCS_Info.Mcptt, -, v_MCS_Info.Mcvideo, -, v_MCS_Info.Mcdata);

      // Store client-id (if any)
      fl_MCX_SIP_Registration_ClientId_Set(v_XmlBodies);
    }
    // Step 4
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_MCX_RegisterResponse_200_MessageHeaderTX(v_RegisterReq, p_ServiceType))));

    return v_IMS_DATA_REQ;
  }



2.2 Runtime with f_MCX_IMS_PTC_SecurityParameters_GetKey
	Function names
	f_MCX_IMS_PTC_SecurityParameters_GetKey,
f_MCX_SecurityParameters_GetKey,
f_MCX_SecurityParameters_GetOptionalKey (new)

	Reason for change
	f_MCX_SecurityParameters_GetKey causes a runtime error when a key does not exist; nevertheless f_MCX_IMS_PTC_SecurityParameters_GetKey can cope with not existing keys

	Summary of change
	f_MCX_SecurityParameters_GetKey renamed to f_MCX_SecurityParameters_GetExistingKey and using new function f_MCX_SecurityParameters_GetOptionalKey

	TTCN modules
	MCX_IMS_Component, MCX_SecurityParameters



	  function f_MCX_IMS_PTC_SecurityParameters_GetKey(MCX_KeyIndex_Type p_KeyIndex) runs on MCX_IMS_PTC return template (omit) MCX_SecurityKey_Type
  {
    var MCX_SecurityParameters_Type v_SecurityParameters;
    var template (omit) MCX_SecurityKey_Type v_Key := omit;

    v_SecurityParameters := f_MCX_IMS_PTC_SecurityParameters_Get();
    v_Key := f_MCX_SecurityParameters_GetOptionalKey(v_SecurityParameters, p_KeyIndex);

    return v_Key;
  }

	  function f_MCX_SecurityParameters_GetExistingKey(MCX_SecurityParameters_Type p_SecurityParameters,
                                           MCX_KeyIndex_Type p_KeyIndex) return MCX_SecurityKey_Type
  {
    var template (omit) MCX_SecurityKey_Type v_Key := f_MCX_SecurityParameters_GetOptionalKey(p_SecurityParameters, p_KeyIndex);

    if (not isvalue(v_Key)) {
      FatalError(__FILE__, __LINE__, "key is not initialised");
    }
    return valueof(v_Key);

    var integer v_Index := enum2int(p_KeyIndex);

    if (not f_MCX_SecurityParameters_KeyExists(p_SecurityParameters, p_KeyIndex)) {
      FatalError(__FILE__, __LINE__, "key is not initialised");
    }
    return p_SecurityParameters.Keys[v_Index].Key;
  }

	  function f_MCX_SecurityParameters_GetOptionalKey(MCX_SecurityParameters_Type p_SecurityParameters,
                                                   MCX_KeyIndex_Type p_KeyIndex) return template (omit) MCX_SecurityKey_Type
  {
    var integer v_Index := enum2int(p_KeyIndex);

    if (not f_MCX_SecurityParameters_KeyExists(p_SecurityParameters, p_KeyIndex)) {
      return omit;
    }
    return p_SecurityParameters.Keys[v_Index].Key;
  }



